
Introduction

CoFI

➤ CoFI is an initiative to provide a common framework for algebraic

specification and development.

Casl

➤ Casl has been designed by CoFI as a general-purpose algebraic

specification language, subsuming many existing languages.

About This Tutorial
➤ This tutorial illustrates and discusses how to write Casl

specifications.

Underlying Concepts

➤ Casl is based on standard concepts of algebraic specification.

Total, Many-Sorted Specifications

➤ Total, many-sorted specifications in Casl may be written essentially

as in many other algebraic specification languages.

➤ Casl provides also useful abbreviations.

➤ Casl allows loose, generated and free specifications.

Loose Specifications

➤ Casl syntax for declarations and axioms involves familiar notation,

and is mostly self-explanatory.

spec Strict Partial Order =

%% Let’s start with a simple example !

sort Elem
pred < : Elem × Elem %% pred abbreviates predicate

∀x , y , z : Elem
• ¬(x < x ) %(strict)%

• x < y ⇒ ¬(y < x ) %(asymmetric)%

• x < y ∧ y < z ⇒ x < z %(transitive)%

%{ Note that there may exist x, y such that
neither x < y nor y < x. }%

end

➤ Specifications can easily be extended by new declarations and axioms.

spec Total Order =

Strict Partial Order

then ∀x , y : Elem • x < y ∨ y < x ∨ x = y %(total)%

ops min(x , y : Elem) : Elem = x when x < y else y ;
max (x , y : Elem) : Elem = y when min(x , y) = x else x

end

%display leq %LATEX ≤
spec Partial Order =

Strict Partial Order

then pred ≤ (x , y : Elem) ⇔ (x < y ∨ x = y)
end

➤ The %implies annotation is used to indicate that some axioms are

supposed to be consequences of others.

spec Partial Order 1 =

Partial Order

then %implies

∀x , y , z : Elem • x ≤ y ∧ y ≤ z ⇒ x ≤ z %(transitive)%

end

➤ Attributes may be used to abbreviate axioms for associativity,

commutativity, idempotence, and unit properties.

spec Monoid =

sort Monoid
ops 1 : Monoid ;

∗ : Monoid ×Monoid → Monoid , assoc, unit 1
end

➤ Genericity of specifications can be made explicit using parameters.

spec Generic Monoid [sort Elem] =

sort Monoid
ops inj : Elem → Monoid ;

1 : Monoid ;
∗ : Monoid ×Monoid → Monoid , assoc, unit 1

∀x , y : Elem • inj (x ) = inj (y) ⇒ x = y
end

➤ References to generic specifications always instantiate the parameters.

spec Generic Commutative Monoid [sort Elem] =

Generic Monoid [sort Elem]

then ∀x , y : Monoid • x ∗ y = y ∗ x
end

spec Generic Commutative Monoid 1 [sort Elem] =

Generic Monoid [sort Elem]

then op ∗ : Monoid ×Monoid → Monoid , comm
end

➤ Datatype declarations may be used to abbreviate declarations of sorts

and constructors.

spec Container [sort Elem] =

type Container ::= empty | insert(Elem; Container)
pred is in : Elem × Container
∀e, e ′ : Elem; C : Container
• ¬(e is in empty)
• e is in insert(e ′, C ) ⇔ (e = e ′ ∨ e is in C )

end

➤ Loose datatype declarations are appropriate when further

constructors may be added in extensions.

spec Marking Container [sort Elem] =

Container [sort Elem]

then type Container ::= mark insert(Elem; Container)
pred is marked in : Elem × Container
∀e, e ′ : Elem; C : Container
• e is in mark insert(e ′, C ) ⇔ (e = e ′ ∨ e is in C )
• ¬(e is marked in empty)
• e is marked in insert(e ′, C ) ⇔ e is marked in C
• e is marked in mark insert(e ′, C ) ⇔

(e = e ′ ∨ e is marked in C )
end

c© Michel Bidoit, Peter D. Mosses, CoFI 16 Casl Tutorial



Generated Specifications

➤ Sorts may be specified as generated by their constructors.

spec Generated Container [sort Elem] =

generated type Container ::= empty | insert(Elem; Container)
pred is in : Elem × Container
∀e, e ′ : Elem; C : Container
• ¬(e is in empty)
• e is in insert(e ′, C ) ⇔ (e = e ′ ∨ e is in C )

end

➤ Generated specifications are in general loose.

spec Generated Container Merge [sort Elem] =

Generated Container [sort Elem]

then op merge : Container × Container → Container
∀e : Elem; C , C ′ : Container
• e is in (C merge C ′) ⇔ (e is in C ∨ e is in C ′)

end

➤ Generated specifications need not be loose.

spec Generated Set [sort Elem] =

generated type Set ::= empty | insert(Elem ; Set)

pred is in : Elem × Set

ops { }(e : Elem) : Set = insert(e, empty);

∪ : Set × Set → Set;

remove : Elem × Set → Set

∀e, e′ : Elem ; S , S ′ : Set

• ¬(e is in empty)

• e is in insert(e′ , S) ⇔ (e = e′ ∨ e is in S)

• S = S ′ ⇔ (∀x : Elem • x is in S ⇔ x is in S ′)
• e is in (S ∪ S ′) ⇔ (e is in S ∨ e is in S ′)
• e is in remove(e′, S) ⇔ (¬(e = e′) ∧ e is in S)

then %implies

generated type Set ::= empty | { }(Elem) | ∪ (Set; Set)

op ∪ : Set × Set → Set, assoc, comm, idem , unit empty

∀e, e′ : Elem ; S : Set

• insert(e, insert(e, S)) = insert(e, S)

• insert(e, insert(e′ , S)) = insert(e′, insert(e, S))
end

➤ Generated types may need to be declared together.

sort Node
generated type Tree ::= mktree(Node; Forest)
generated type Forest ::= empty | add(Tree; Forest)

is both incorrect (linear visibility) and wrong (the corresponding

semantics is not the “expected” one). One must write instead:

sort Node
generated types Tree ::= mktree(Node; Forest);

Forest ::= empty | add(Tree; Forest)

Free Specifications

➤ Free specifications provide initial semantics and avoid the need for

explicit negation.

spec Natural = free type Nat ::= 0 | suc(Nat) end

➤ Free datatype declarations are particularly convenient for defining

enumerated datatypes.

spec Color =

free type RGB ::= Red | Green | Blue
free type Printer Color ::= Cyan | Magenta | Yellow | Black

end

➤ Free specifications can also be used when the constructors are related

by some axioms.

spec Integer =

free { type Int ::= 0 | suc(Int) | pre(Int)
∀x , y : Int
• suc(pre(x )) = x
• pre(suc(x )) = x }

end

➤ Predicates hold minimally in models of free specifications.

spec Natural Order =

Natural

then free { pred < : Nat ×Nat
∀x , y : Nat
• 0 < suc(x )
• x < y ⇒ suc(x ) < suc(y) }

end

➤ Operations may be safely defined by induction on the constructors of

a free datatype declaration.

spec Natural Arithmetic =

Natural Order

then ops 1 : Nat = suc(0 );
+ : Nat × Nat → Nat , assoc, comm, unit 0 ;
∗ : Nat × Nat → Nat , assoc, comm, unit 1

∀x , y : Nat
• x + suc(y) = suc(x + y)
• x ∗ 0 = 0
• x ∗ suc(y) = x + (x ∗ y)

end

➤ More care may be needed when defining operations on free datatypes

with axioms relating the constructors.

spec Integer Arithmetic =

Integer

then ops 1 : Int = suc(0 );

+ : Int × Int → Int, assoc, comm, unit 0 ;

− : Int × Int → Int;

∗ : Int × Int → Int, assoc, comm, unit 1

∀x , y : Int

• x + suc(y) = suc(x + y)

• x + pre(y) = pre(x + y)

• x − 0 = x

• x − suc(y) = pre(x − y)

• x − pre(y) = suc(x − y)

• x ∗ 0 = 0

• x ∗ suc(y) = (x ∗ y) + x

• x ∗ pre(y) = (x ∗ y)− x
end

spec Integer Arithmetic Order =

Integer Arithmetic

then preds ≤ , ≥ : Int × Int
∀x , y : Int
• 0 ≤ 0
• ¬(0 ≤ pre(0 ))
• 0 ≤ x ⇒ 0 ≤ suc(x )
• ¬(0 ≤ x ) ⇒ ¬(0 ≤ pre(x ))
• suc(x ) ≤ y ⇔ x ≤ pre(y)
• pre(x ) ≤ y ⇔ x ≤ suc(y)
• x ≥ y ⇔ y ≤ x

end

➤ Generic specifications often involve free extensions of (loose)

parameters.

spec List [sort Elem] =

free type List ::= empty | cons(Elem; List)
end

spec Set [sort Elem] =

free { type Set ::= empty | insert(Elem; Set)
pred is in : Elem × Set
∀e, e ′ : Elem; S : Set
• insert(e, insert(e, S )) = insert(e, S )
• insert(e, insert(e ′, S )) = insert(e ′, insert(e, S ))
• ¬(e is in empty)
• e is in insert(e, S )
• e is in insert(e ′, S ) if e is in S }

end

spec Transitive Closure [sort Elem pred R : Elem × Elem] =

free { pred R+ : Elem × Elem
∀x , y , z : Elem
• x R y ⇒ x R+y
• x R+y ∧ y R+z ⇒ x R+z }

end

➤ Loose extensions of free specifications can avoid overspecification.

spec Natural Bounded =

Natural Arithmetic

then op max size : Nat axiom 0 < max size end

spec Set Choose [sort Elem] =

Set [sort Elem]

then op choose : Set → Elem
∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S

end

➤ Datatypes with observer operations can be specified as generated

instead of free.

spec Set Generated [sort Elem] =

generated type Set ::= empty | insert(Elem; Set)
pred is in : Elem × Set
∀e, e ′ : Elem; S , S ′ : Set
• ¬(e is in empty)
• e is in insert(e ′, S ) ⇔ (e = e ′ ∨ e is in S )
• S = S ′ ⇔ (∀x : Elem • x is in S ⇔ x is in S ′)

end

➤ The %def annotation is useful to indicate that some operations or

predicates are uniquely defined.

spec Set Union [sort Elem] =

Set [sort Elem]

then %def

ops ∪ : Set × Set → Set , assoc, comm, idem, unit empty ;
remove : Elem × Set → Set

∀e, e ′ : Elem; S , S ′ : Set
• S ∪ insert(e ′, S ′) = insert(e ′, S ∪ S ′)
• remove(e, empty) = empty
• remove(e, insert(e, S )) = remove(e, S )
• remove(e, insert(e ′, S )) =

insert(e ′, remove(e, S )) if ¬(e = e ′)
end

➤ Operations can be defined by axioms involving observer operations,

instead of inductively on constructors.

spec Set Union 1 [sort Elem] =

Set Generated [sort Elem]

then %def

ops ∪ : Set × Set → Set , assoc, comm, idem, unit empty ;
remove : Elem × Set → Set

∀e, e ′ : Elem; S , S ′ : Set
• e is in (S ∪ S ′) ⇔ (e is in S ∨ e is in S ′)
• e is in remove(e ′, S ) ⇔ (¬(e = e ′) ∧ e is in S )

end

➤ Sorts declared in free specifications are not necessarily generated by

their constructors.

spec Number =

free { type Number ::= 0 | suc(Number)
op + : Number × Number → Number ,

assoc, comm, unit 0
∀x , y : Number • x + suc(y) = suc(x + y)
∀x : Number • ∃y : Number • x + y = 0 }

end

Partial Functions

➤ Partial functions arise naturally. They are declared differently from

total functions.

spec Set Partial Choose [sort Elem] =

Generated Set [sort Elem]

then op choose : Set →? Elem
end

➤ Terms containing partial functions may be undefined, i.e., they may

not denote any value.

E.g., the term choose(empty) may be undefined.

➤ Functions, even total ones, propagate undefinedness.

If the term choose(S ) is undefined, then the term insert(choose(S ), S ′)
is undefined as well, although insert is a total function.

➤ Predicates applied to an undefined term yield false.

If the term choose(S ) is undefined, then choose(S ) is in S yields false.

➤ Equations hold also when both terms are undefined.

The ordinary equation

insert(choose(S ), insert(choose(S ), empty))
= insert(choose(S ), empty)

holds also when the term choose(S ) is undefined.

➤ Special care is needed in specifications involving partial functions.

Asserting choose(S ) is in S as an axiom implies that choose(S ) is

defined, for any S .

Asserting remove(choose(S ), insert(choose(S ), empty)) = empty as an

axiom implies that choose(S ) is defined for any S , since the term empty
is always defined.

Asserting insert(choose(S ), S ) = S as an axiom implies that choose(S )
is defined for any S , since a variable always denotes a defined value.

➤ The definedness of a term can be checked or asserted.

spec Set Partial Choose 1 [sort Elem] =

Set Partial Choose [sort Elem]

then • ¬(def choose(empty))
∀S : Set • def choose(S ) ⇒ choose(S ) is in S

end

We know that choose is undefined on empty , but we don’t know exactly

when choose(S ) is defined (it may be undefined on other values than

empty).

If we would have specified choose by:

∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S
then we could conclude that choose(S ) is defined when S is not equal to

empty , but nothing about the undefinedness of choose(empty).

➤ The domains of definition of partial functions can be specified exactly.

spec Set Partial Choose 2 [sort Elem] =

Set Partial Choose [sort Elem]

then ∀S : Set • def choose(S ) ⇔ ¬(S = empty)
∀S : Set • def choose(S ) ⇒ choose(S ) is in S

end

➤ Loosely specified domains of definition may be useful.

spec Natural Bounded Addition =

Natural Bounded

then op +? : Nat × Nat →? Nat
∀x , y : Nat
• def (x+?y) if x + y < max size

%{ x + y < max size implies both

x < max size and y < max size }%
• def (x+?y) ⇒ x+?y = x + y

end

c© Michel Bidoit, Peter D. Mosses, CoFI 47 Casl Tutorial

➤ Domains of definition can be specified more or less explicitly.

spec Set Partial Choose 3 [sort Elem] =

Set Partial Choose [sort Elem]

then • ¬(def choose(empty))
∀S : Set • ¬(S = empty) ⇒ choose(S ) is in S

end

We can conclude after some reasoning that

def choose(S ) ⇔ ¬(S = empty), but this is not so prominent.

spec Natural Partial Pre =

Natural Arithmetic

then op pre : Nat →? Nat
∀x , y : Nat
• ¬ def pre(0 )
• pre(suc(x )) = x

end

is explicit enough.

spec Natural Partial Subtraction =

Natural Partial Pre

then op − : Nat × Nat →? Nat
∀x , y : Nat
• x − 0 = x
• x − suc(y) = pre(x − y)

end

is correct, but clearly not explicit enough, and better specified as follows:

spec Natural Partial Subtraction 1 =

Natural Partial Pre

then op − : Nat × Nat →? Nat
∀x , y : Nat
• def (x − y) ⇔ (y < x ∨ y = x )
• x − 0 = x
• x − suc(y) = pre(x − y)

end

➤ Partial functions are minimally defined by default in free

specifications.

spec List Selectors [sort Elem] =

List [sort Elem]

then free { ops head : List →? Elem;
tail : List →? List

∀e : Elem; L : List
• head(cons(e, L)) = e
• tail(cons(e, L)) = L }

end

spec List Selectors 1 [sort Elem] =

List [sort Elem]

then ops head : List →? Elem;
tail : List →? List

∀e : Elem; L : List
• ¬ def head(nil)
• ¬ def tail(nil)
• head(cons(e, L)) = e
• tail(cons(e, L)) = L

end

c© Michel Bidoit, Peter D. Mosses, CoFI 52 Casl Tutorial



➤ Selectors can be specified concisely in datatype declarations, and are

usually partial.

spec List Selectors 2 [sort Elem] =

free type List ::= nil | cons(head :? Elem; tail :? List)
end

spec Natural Suc Pre =

free type Nat ::= 0 | suc(pre :? Nat)
. . .

end

➤ Selectors are usually total when there is only one constructor.

spec Pair [sort Elem1 ] [sort Elem2 ] =

free type Pair ::= pair(first : Elem1 ; second : Elem2 )
end

➤ Constructors also may be partial.

spec Part Container [sort Elem] =

generated type

P Container ::= empty | insert(Elem; P Container)?
pred addable : Elem × P Container
vars e, e ′ : Elem; C : P Container
• def insert(e, C ) ⇔ addable(e, C )
pred is in : Elem × P Container
• ¬(e is in empty)
• (e is in insert(e ′, C ) ⇔ (e = e ′ ∨ e is in C )) if addable(e ′, C )

end

➤ Existential equality involves the definedness of both terms as well as

their equality.

spec Natural Partial Subtraction 2 =

Natural Partial Subtraction 1

then

∀x , y : Nat
%{ y − x = z − x ⇒ y = z would be wrong,

def (y − x) ∧ def (z − x) ∧ y − x = z − x ⇒ y = z

is correct, but can be better abbreviated into: }%
• y − x e= z − x ⇒ y = z

end

Subsorting

➤ Sorts may denote related domains, which can be expressed using

subsorting.

spec Generic Monoid 1 [sort Elem] =

sorts Elem < Monoid
ops 1 : Monoid ;

∗ : Monoid ×Monoid → Monoid , assoc, unit 1
end

➤ A sort may be essentially the union of its subsorts.

spec Vehicle =

Natural

then sorts Car , Bicycle < Vehicle
ops max speed : Vehicle → Nat ;

weight : Vehicle → Nat ;
engine capacity : Car → Nat

end

c© Michel Bidoit, Peter D. Mosses, CoFI 59 Casl Tutorial

➤ Operations declared on some sort are automatically inherited by its

subsorts, even those declared later on.

spec More Vehicle = Vehicle then sorts Boat < Vehicle end

c© Michel Bidoit, Peter D. Mosses, CoFI 60 Casl Tutorial



➤ Subsort membership can be checked or asserted.

spec Speed Regulation =

Vehicle

then op speed limit : Vehicle → Nat
∀v : Vehicle
• v ∈ Car ⇒ speed limit(v) = 130
• v ∈ Bicycle ⇒ speed limit(v) = 30

end

➤ Datatype declarations can involve subsort declarations.

type Vehicle ::= sort Car | sort Bicycle | sort Boat
leaves the way open to further kinds of vehicles (e.g., planes).

generated type Vehicle ::= sort Car | sort Bicycle | sort Boat
prevents the definition of further subsorts, e.g., for planes.

free type Vehicle ::= sort Car | sort Bicycle | sort Boat
prevents the definition of further subsorts, and moreover the definition of

a common subsort of both Car and Boat
(e.g., sorts Amphibious < Car , Boat).

➤ Subsorts may also arise as classifications of previously specified

values, and their values can be explicitly defined.

spec Natural Subsorts =

Natural Arithmetic

then pred even : Nat
• even(0 )
• ¬ even(1 )
∀n : Nat • even(suc(suc(n)))⇔ even(n)
sort Even = {x : Nat • even(x )}
sort Prime = {x : Nat •

∀y , z : Nat • x = y ∗ z ⇒ y = 1 ∨ z = 1}
end

spec Positive =

Natural Partial Pre

then sort Pos = {x : Nat • ¬(x = 0 )} end

➤ It may be useful to redeclare previously defined operations, using the

new subsorts introduced.

spec Positive Arithmetic =

Positive

then ops 1 : Pos;
suc : Nat → Pos;

+ , ∗ : Pos × Pos → Pos;
+ : Pos × Nat → Pos;
+ : Nat × Pos → Pos

end

➤ A subsort may correspond to the definition domain of a partial

function.

spec Positive Pre = Positive Arithmetic then

op pre : Pos → Nat
end

➤ Using subsorts may avoid the need for partial functions.

spec Natural Positive Arithmetic =

free types Nat ::= 0 | sort Pos;
Pos ::= suc(pre : Nat)

ops 1 : Pos = suc(0 );
+ : Nat × Nat → Nat , assoc, comm, unit 0 ;
∗ : Nat × Nat → Nat , assoc, comm, unit 1 ;
+ , ∗ : Pos × Pos → Pos;
+ : Pos × Nat → Pos;
+ : Nat × Pos → Pos

∀x , y : Nat
• x + suc(y) = suc(x + y)
• x ∗ 0 = 0
• x ∗ suc(y) = x + (x ∗ y)

end

➤ Casting a term from a supersort to a subsort is explicit and may be

undefined.

pre( pre(suc(1 )) as Pos )

def pre( pre(suc(1 )) as Pos )

¬ def ( pre( pre(suc(1 )) as Pos ) as Pos )

➤ Supersorts may be useful when generalizing previously specified sorts.

spec Integer Arithmetic 1 =

Natural Positive Arithmetic

then free type Int ::= sort Nat | − (Pos)

ops + : Int × Int → Int, assoc, comm, unit 0 ;

− : Int × Int → Int;

∗ : Int × Int → Int, assoc, comm, unit 1 ;

| | : Int → Nat

∀x , y : Int; n : Nat; p, q : Pos

• suc(n) + (−1 ) = n

• suc(n) + (−suc(q)) = n + (−q)

• (−p) + (−q) = −(p + q)

• x − 0 = x

• x − p = x + (−p)

• x − (−q) = x + q

• n ∗ (−q) = −(n ∗ q)

• (−p) ∗ (−q) = p ∗ q

• |n| = n

• | − p| = p
end

➤ Supersorts may also be used for extending the intended values by

new values representing errors or exceptions.

spec Set Error Choose [sort Elem] =

Generated Set [sort Elem]

then sorts Elem < ElemError

op choose : Set → ElemError

pred is in : ElemError × Set

∀S : Set • ¬(S = empty) ⇒ choose(S) ∈ Elem ∧
choose(S) is in S

end

spec Set Error Choose 1 [sort Elem] =

Generated Set [sort Elem]

then sorts Elem < ElemError

op choose : Set → ElemError

∀S : Set • ¬(S = empty) ⇒ choose(S) ∈ Elem ∧
(choose(S) as Elem) is in S

end

Structuring Specifications

➤ Union and extension can be used to structure specifications.

spec List Set [sort Elem] =

List Selectors 2 [sort Elem] and

Generated Set [sort Elem]

then op elements of : List → Set
∀e : Elem; L : List
• elements of nil = empty
• elements of cons(e, L) = {e} ∪ elements of L

end

➤ Arbitrary parts of specifications can have initial semantics.

spec List Choose [sort Elem] =

List Selectors 2 [sort Elem] and Set Partial Choose 2 [sort Elem]

then ops elements of : List → Set;

choose : List →? Elem

∀e : Elem ; L : List

• elements of nil = empty

• elements of cons(e, L) = {e} ∪ elements of L

• def choose(L) ⇔ ¬(L = nil)

• choose(L) = choose(elements of L)
end

spec Set to List [sort Elem] =

List Set [sort Elem]

then op list of : Set → List

∀S : Set • elements of (list of S) = S

end

➤ Renaming may be used to avoid unintended name clashes, or to

adjust names of sorts and change notations for operations and

predicates.

spec Stack [sort Elem] =

List Selectors 2 [sort Elem] with sort List 7→ Stack ,

ops cons 7→ push onto ,

head 7→ top,

tail 7→ pop
end

➤ When combining specifications, origins of symbols can be indicated.

spec List Set 1 [sort Elem] =

List Selectors 2 [sort Elem] with nil , cons
and Generated Set [sort Elem] with empty , { }, ∪
then op elements of : List → Set

∀e : Elem; L : List
• elements of nil = empty
• elements of cons(e, L) = {e} ∪ elements of L

end

c© Michel Bidoit, Peter D. Mosses, CoFI 74 Casl Tutorial



➤ Auxiliary symbols used in structured specifications can be hidden.

spec Natural Partial Subtraction 3 =

Natural Partial Subtraction 1 hide suc, pre end

spec Natural Partial Subtraction 4 =

Natural Partial Subtraction 1

reveal Nat , 0 , 1 , + , − , ∗ , <

end

➤ Auxiliary symbols can be made local when they do not need to be

exported.

spec List Order [Partial Order with sort Elem, pred ≤ ] =

List Selectors 2 [sort Elem]

then local op insert : Elem × List → List
∀e, e ′ : Elem; L : List
• insert(e, nil) = cons(e, nil)
• insert(e, cons(e ′, L)) = cons(e, cons(e ′, L)) when e ≤ e ′

else cons(e ′, insert(e, L))
within op order : List → List

∀e : Elem; L : List
• order(nil) = nil
• order(cons(e, L)) = insert(e, order(L))

end

spec List Order Sorted [Partial Order with sort Elem , pred ≤ ] =

List Selectors 2 [sort Elem]

then local preds is sorted : List;

is in : Elem × List;

same elements : List × List

∀e, e′ : Elem ; L, L′ : List

• nil is sorted

• cons(e, nil) is sorted

• cons(e, cons(e′,L)) is sorted ⇔
(e ≤ e′ ∧ cons(e′, L) is sorted)

• ¬(e is in nil)

• e is in cons(e′, L) ⇔ (e = e′ ∨ e is in L)

• same elements(L, L′) ⇔
(∀e : Elem • e is in L ⇔ e is in L′)

within op order : List → List

∀L : List • order(L) is sorted ∧
same elements(L, order(L))

end

c© Michel Bidoit, Peter D. Mosses, CoFI 77 Casl Tutorial

Generic Specifications

➤ Explicit parameters show the intended genericity of a specification.

➤ Parameters are arbitrary specifications.

spec List Selectors 2 [sort Elem] = . . .

spec List Order [Partial Order with sort Elem, pred ≤ ] =

. . .

➤ A generic specification may have more than one parameter.

spec Pair [sort Elem1 ] [sort Elem2 ] = . . .

spec Index =

Total Order with Elem 7→ Index hide min, max

then ops first, last : Index

• ¬(last < first)

end

spec Array [Index] [sort Elem] =

type Array ::= Init | [ ]:= (Array ; Index ; Elem)?

op [ ] : Array × Index →? Elem

then local pred is valid : Index

∀i : Index

• i is valid ⇔ ¬(i < first) ∧ ¬(last < i)

within ∀T : Array ; i , j : Index ; e, e′ : Elem

• def T [i ]:=e ⇔ i is valid

• def T [i ] ⇔ (i is valid ∧
∃T ′ : Array ; e : Elem • T = T ′[i ]:=e)

• def T [i ]:=e ⇒ (T [i ]:=e)[i ] = e

• ¬(i = j ) ∧ def T [i ]:=e ∧ def T [j ] ⇒
(T [i ]:=e)[j ] = T [j ]

• def T [i ]:=e ⇒ (T [i ]:=e)[i ]:=e′ = T [i ]:=e′
• ¬(i = j ) ∧ def T [i ]:=e ∧ def T [j ]:=e′ ⇒

(T [i ]:=e)[j ]:=e′ = (T [j ]:=e′)[i ]:=e
end

➤ In instantiations, the intended fitting of the parameter symbols to the

argument symbols has to be uniquely determined.

spec List Integer =

List Selectors 2 [Integer Arithmetic Order fit Elem 7→ Int]

end

spec Pair Natural =

Pair [Natural Bounded fit Elem1 7→ Nat]

[Natural Bounded fit Elem2 7→ Nat]

end

Trivial instantiations can be concisely described by omitting the fitting

symbol map.

Note that an alternative to Pair Natural is to write:

spec Pair Natural 1 = Pair [sort Nat] [sort Nat] and Natural Bounded end

➤ Implied and identity mappings can be omitted.

spec Array Indexed by Nat [sort Elem] =

Array [Natural Bounded

fit Index 7→ Nat , first 7→ 0 , last 7→ max size]

[sort Elem]

end

Note that there is no need to explicitly map < . . .

➤ Composition of generic specifications is expressed using instantiation.

spec Set List [sort Elem] =

Generated Set

[List Selectors 2 [sort Elem] fit Elem 7→ List ]
end

➤ Compound sorts introduced by a generic specification get

automatically renamed on instantiation, which avoids name clashes.

spec List 1 [sort Elem] =

free type List [Elem] ::= nil |
cons(head :? Elem; tail :? List [Elem])

ops ++ : List [Elem]× List [Elem] → List [Elem],
assoc, unit nil ;

reverse : List [Elem] → List [Elem]
∀e : Elem; L, L1 , L2 : List [Elem]
• cons(e, L1 ) ++ L2 = cons(e, L1 ++ L2 )
• reverse(nil) = nil
• reverse(cons(e, L)) = reverse(L) ++ cons(e, nil)

end

spec Two Lists =

List 1 [Natural Arithmetic fit Elem 7→ Nat ]
%% Provides sort List[Nat]

and List 1 [Integer Arithmetic Order fit Elem 7→ Int ]
%% Provides sort List[Int]

end

spec Two Lists 1 =

List 1 [Integer Arithmetic 1 fit Elem 7→ Nat ]
and List 1 [Integer Arithmetic 1 fit Elem 7→ Int ]
end

➤ Compound symbols can also be used for operations and predicates.

spec List Order Sorted 1 [Partial Order with sort Elem , pred ≤ ] =

List 1 [sort Elem]

then local op insert : Elem × List[Elem ] → List[Elem ]

∀e, e′ : Elem ; L : List[Elem ]

• insert(e, nil) = cons(e, nil)

• insert(e, cons(e′, L)) = cons(e, cons(e′, L)) when e ≤ e′
else cons(e′, insert(e, L))

within op order [ ≤ ] : List[Elem ] → List[Elem ]

∀e : Elem ; L : List[Elem ]

• order [ ≤ ](nil) = nil

• order [ ≤ ](cons(e,L)) = insert(e, order [ ≤ ](L))
end

spec List Reverse Orders =

List Order Sorted 1

[Integer Arithmetic Order fit Elem 7→ Int, ≤ 7→ ≤ ]

and List Order Sorted 1

[Integer Arithmetic Order fit Elem 7→ Int, ≤ 7→ ≥ ]

then %implies

∀L : List[Int] • order [ ≤ ](L) = reverse(order [ ≥ ](L))

end

➤ Parameters should be distinguished from references to fixed

specifications that are not intended to be instantiated.

spec List Length [sort Elem] given Natural Arithmetic =

List 1 [sort Elem]

then op length : List[Elem ] → Nat

∀e : Elem ; L : List[Elem ]

• length(nil) = 0

• length(cons(e, L)) = length(L) + 1

then %implies

∀L : List[Elem ] • length(reverse(L)) = length(L)

end

spec List Natural = List Length [Natural Arithmetic fit Elem 7→ Nat]

spec Wrong List Length [sort Elem] =

Natural Arithmetic and List 1 [sort Elem]

then . . . end

would be inadequate since instantiation by Natural Arithmetic is ill-formed.

➤ Views are named fitting maps, and can be defined along with

specifications.

view Integer as Partial Order 1 :

Partial Order to Integer Arithmetic Order =

Elem 7→ Int, ≤ 7→ ≤

view Integer as Partial Order 2 :

Partial Order to Integer Arithmetic Order =

Elem 7→ Int, ≤ 7→ ≥

spec List Reverse Orders 1 =

List Order Sorted 1 [view Integer as Partial Order 1]

and List Order Sorted 1 [view Integer as Partial Order 2]

then %implies ∀L : List[Int] • order [ ≤ ](L) = reverse(order [ ≥ ](L))

end

➤ Views can also be generic.

Specifying Architectural Structure

Libraries

